

Définition de syntaxes concrètes textuelles des DSLs (Xtext)

La version d'Eclipse à utiliser est la suivante (Eclipse EMF version 2020-09 (4.17.0), installée sur les machines de l'N7) :

`/mnt/n7fs/ens/tp_ouederni/eclipse-idm/eclipse`

La syntaxe abstraite d'un DSML (Domain Specific Modelling Language, par exemple SimplePDL. ecore dans notre cas et qui est exprimée en Ecore, autre langage de métamodélisation est possible) ne peut pas être manipulée directement. Le projet EMF d'Eclipse permet d'engendrer un éditeur arborescent et les classes Java qui permettront de manipuler un modèle conforme à un métamodèle. Cependant, ce ne sont pas des outils très pratiques lorsque l'on veut saisir un modèle. Aussi, il est souhaitable d'associer à une syntaxe abstraite une ou plusieurs syntaxes concrètes pour faciliter la construction et la modification des modèles. Ces syntaxes concrètes peuvent être textuelles (l'objet de cette séance) ou graphiques (séance précédente).

Pour la définition des syntaxes concrètes textuelles, nous allons utiliser l'outil *Xtext*¹ de openArchitectureWare (oAW). *Xtext* fait partie du projet TMF (Textual Modeling Framework). Il permet non seulement de définir une syntaxe textuelle pour un DSL mais aussi de disposer, au travers d'Eclipse, d'un environnement de développement pour ce langage, avec en particulier un éditeur syntaxique (coloration, complétion, outline, détection et visualisation des erreurs, etc).

Xtext s'appuie sur un générateur d'analyseurs descendants récursifs (LL(k)). L'idée est de décrire à la fois la syntaxe concrète du langage et comment le modèle EMF est construit en mémoire au fur et à mesure de l'analyse syntaxique.

1 Définir des syntaxes concrètes textuelles pour SimplePDL

Xtext consiste à définir une syntaxe concrète textuelle pour un DSML au travers d'une grammaire qui doit pouvoir être analysée en LL(k). En particulier, elle ne doit pas être récursive à gauche² ! Généralement, *Xtext* est utilisé pour engendrer le métamodèle correspondant. C'est ce que nous allons voir dans ce sujet. *Xtext* peut aussi travailler à partir d'un métamodèle existant. Cependant, il ne faut pas que la distance soit trop grande entre le métamodèle existant et la syntaxe textuelle souhaitée sinon il est conseillé de définir un métamodèle adapté à la syntaxe

1. <https://www.eclipse.org/Xtext/>

2. Voir la documentation de *Xtext* : <http://www.eclipse.org/Xtext/documentation>

concrète (par exemple engendré automatiquement) puis d'écrire une transformation de modèle à modèle pour obtenir un modèle conforme au métamodèle existant du DSML.

Exercice 1 : Première syntaxe textuelle pour SimplePDL : PDL1

Comme toujours avec Eclipse, il faut commencer par créer un nouveau projet.

1.1 Création du projet Xtext. Sous Eclipse, on commence par définir un projet (New > Project, puis Xtext > Xtext project). Il faut indiquer d'abord le nom du projet principal (fr.n7.pdl1 par exemple³), le nom du langage (fr.n7.PDL1), et l'extension pour les fichiers contenant les modèles correspondants (pd11).

Plusieurs projets sont alors créés, en particulier fr.n7.pdl1 qui contient dans src le fichier *Xtext* qui décrit la syntaxe de PDL1. Ce fichier nommé pd11.xtext apparaît d'ailleurs dans l'éditeur. Il contient un squelette montrant la structure d'un fichier *Xtext*. La première ligne indique le langage qui est défini. La ligne suivante indique le métamodèle qui sera engendré. Le reste du fichier est composé des règles de production définissant la syntaxe concrète textuelle avec les actions qui permettent de construire le modèle EMF correspondant.

1.2 Définition de la syntaxe PDL1. Un exemple de la syntaxe textuelle souhaitée est donné dans l'exemple ci-dessous. Le fichier *Xtext* correspondant est donné au listing 1.

```
process dvp {  
    wd a  
    wd b  
    wd c  
    ws s2s from a to b  
    ws f2f from b to c  
}
```

1.2.1 Expliquer les différents éléments du fichier *Xtext* du listing 1.

1.2.2 Remplacer le texte de PDL1.xtext par celui du listing 1.

1.3 Engendrer l'outillage Xtext pour PDL1. Dans le répertoire src du projet principal, il faut sélectionner le fichier GeneratePDL1.mwe2, cliquer à droite et sélectionner *Run As > MWE Workflow*. La génération prend un peu de temps...

1.4 Utiliser l'éditeur PDL1 engendré. Il faut déployer les greffons *Xtext* engendrés. Ensuite, on peut créer dans le projet fr.n7.simplepdl.exemple un fichier avec l'extension .pd11, par exemple process1.pd11. Saisir un exemple de processus et regarder les possibilités de l'éditeur (colorisation, complétion syntaxique...).

Attention : Le projet doit avoir la nature Xtext. On l'ajoute soit avec clic droit puis *Configure / Convert to Xtext Project*, soit avec un clic droit sur le fichier process1.pd11, *Open with ...* puis *PDL1 Editor* (répondre oui quand Eclipse demande de convertir le projet en un projet Xtext).

1.5 Consulter le métamodèle PDL1 engendré. Le métamodèle PDL1 a été engendré. Le fichier PDL1.ecore est présent dans le répertoire model/generated du projet principal fr.n7.pdl1. Pour l'ouvrir avec l'éditeur graphique, cliquer à droite et sélectionner *Initialize Ecore Diagram*

3. Le nom du projet doit être en minuscules et contenir au moins un point « . ».

Listing 1 – Description *Xtext* pour la syntaxe PDL1

```

1 grammar fr.n7.PDL1 with org.eclipse.xtext.common.Terminals
2 generate pDL1 "http://www.n7.fr/PDL1"
3
4 Process : 'process' name=ID '{'
5           processElements+=ProcessElement*
6           '}' ;
7
8 ProcessElement : WorkDefinition | WorkSequence | Guidance ;
9
10 WorkDefinition : 'wd' name=ID ;
11
12 WorkSequence : 'ws' linkType=WorkSequenceType
13               'from' predecessor=[WorkDefinition]
14               'to' successor=[WorkDefinition] ;
15
16 enum WorkSequenceType : start2start = 's2s'
17                       | finish2start = 'f2s'
18                       | start2finish = 's2f'
19                       | finish2finish = 'f2f' ;
20
21 Guidance : 'note' texte=STRING ;

```

Exercice 2 : Autre syntaxe textuelle pour SimplePDL : PDL2

On considère une nouvelle syntaxe textuelle dont le fichier *Xtext* est donné au listing 2.

2.1 Donner la syntaxe PDL2 du processus donné à la question 1.2.

2.2 Engendrer l'éditeur pour PDL2, le déployer et l'utiliser.

2.3 Comparer les deux métamodèles, PDL1.ecore et PDL2.ecore.

Exercice 3 : Définir une syntaxe textuelle pour un métamodèle existant

Voyons comment utiliser un métamodèle existant plutôt que de laisser *Xtext* l'engendrer. Il faut alors préciser pour chaque règle de grammaire l'élément du méta-modèle qui sera créé.

3.1 S'assurer que les greffons créés à partir du méta-modèle SimplePDL.ecore et du modèle de génération SimplePDL.genmodel ont été déployés.

3.2 Créer un projet avec *New / Project...* et choisir *Xtext Project From Existing Ecore Models*.

Sur le dialogue « Select EPackages », utiliser *Add...* pour sélectionner SimplePDL.genmodel et choisir Process-simplepdl pour le champ *Entry rule*. Faire *Next*.

On arrive sur l'écran classique du projet Xtext où l'on peut renseigner le nom du projet (fr.n7.simplepdl.txt.pdl), le nom du langage (fr.n7.simplepdl.txt.PDL) et l'extension (pd1).

3.3 Lire et comprendre le fichier PDL.xtext produit.

3.4 Exécuter le fichier GeneratePDL.mwe2, puis déployer les greffons engendrés. Utiliser l'éditeur en créant des fichiers avec l'extension .pd1.

3.5 Modifier la grammaire pour reproduire la syntaxe de PDL1.

Listing 2 – Description *Xtext* pour la syntaxe PDL2

```
1 grammar fr.n7.PDL2 with org.eclipse.xtext.common.Terminals
2
3 generate pDL2 "http://www.n7.fr/PDL2"
4
5 Process :
6     'process' name=ID '{'
7         processElements+=ProcessElement*
8     '}' ;
9
10 ProcessElement :
11     WorkDefinition | Guidance ;
12
13 WorkDefinition :
14     'wd' name=ID '{'
15         ( 'starts' 'if' (linksToPredecessors+=DependanceStart)+ )?
16         ( 'finishes' 'if' (linksToPredecessors+=DependanceFinish)+ )?
17     '}' ;
18
19 DependanceStart :
20     predecessor=[WorkDefinition] link=WorkSequenceKindStart ;
21
22 WorkSequenceKindStart:
23     Started2Start = 'started' | Started2Finish = 'finished' ;
24
25 DependanceFinish :
26     predecessor=[WorkDefinition] link=WorkSequenceKindFinish ;
27
28 WorkSequenceKindFinish:
29     Finished2Start = 'started' | Finished2Finish = 'finished' ;
30
31 Guidance :
32     'note' texte=STRING ;
```

2 Application aux réseaux de Petri

Exercice 4 : Définir des syntaxes textuelles pour les réseaux de Petri

Proposer une ou plusieurs syntaxes textuelles pour les réseaux de Petri et utiliser *Xtext* pour engendrer l'outillage Eclipse correspondant.